

User Documentation for SAS Program to Create Measure of Treatment Resistant Depression Using Administrative Claims Data

Version: May 11, 2010

Patients diagnosed with unipolar depression who do not experience symptom relief or remission after initial antidepressant treatment have lower odds of responding to subsequent treatments.¹ The ability to detect early indications of treatment resistant depression (TRD) may facilitate targeting interventions for TRD, as well as managing healthcare costs for patients with TRD. This document describes a SAS program that can be applied to administrative claims data in order to identify patients who are likely to have TRD and measure the degree of treatment resistance. The utility of the algorithm for your patient population depends upon the claims data available, the target population, and utilization patterns of interest, among other factors.

The rest of this document outlines the MGH-AD scoring methodology and the SAS program, including user inputs and outputs of the program.

MGH Staging Method:

The Massachusetts General Hospital (MGH) staging method² assigns points to patients for adequate antidepressant trials, optimization strategies, and electroconvulsive therapy (ECT); the resulting scores indicate the degree of treatment resistance. Patients with lower scores are likely to be non-resistant to treatment (or may be in the early stages of TRD). Patients with intermediate scores are likely to be moderately resistant to treatment (or may be in the early stages of TRD). Patients with high scores are likely to be very resistant to treatment. The table below explains the scoring for the MGH staging method.

Table 1. Calculation of the MGH Staging Method²

Programs	MGH Component	Points	Claims-based definition
MGH	Any use of ECT	3 points	Any claim with a procedure code indicating ECT
MGH MGH_M MGH_AD	Adequate antidepressant trial	1 point	At least 2 fills for the same medication
MGH MGH_M	Optimization strategies (within each adequate trial)		
MGH MGH_M MGH_AD	Extended duration	0.5 point	At least 3 fills for the same medication
MGH MGH_M MGH_AD	Upward titration in dose	0.5 point	Prescription fill for same medication at a higher dose
MGH MGH_M	Augmentation	0.5 point	Prescription for stimulant, atypical antipsychotic, or mood stabilizer initiated during an antidepressant trial



The program (MGH_AD.sas) creates an antidepressant-only version of the score, based on outpatient pharmacy claims only for antidepressant medications. It produces a score (MGH_AD) for each patient and includes patient-level values for each component of the score. Note that, in the study that created this program, the program exhibited sensitivity to the length of time available in the claims data, as 24 months of claims data from the first antidepressant fill were found to be necessary to observe patients with TRD. However, the program is applicable to data spanning shorter time periods.

The output of the antidepressant-only MGH staging method is a patient-level score indicating the degree of treatment resistance. The score ranges from a minimum of zero to an unlimited maximum. The study for which this program was developed found that two years of data were necessary in order to identify higher scores. Scores of 1.5 – 2.0 were the most common. For scores of 3.5 and above (the threshold used for TRD), the percentage of patients at each score declined as the score increased. Due to the small percentage of patients with scores of 6.5 and above, these scores were collapsed into a single category indicating the highest level of treatment resistance.^{3,4}

The program (MGH_AD.sas) must be modified to indicate the SAS library in which the input files are stored. The following code may be found at the beginning of each program: This program requires the following user inputs (described on the next pages): library to store SAS programs, input files, and look-up tables; _pat_file; _ndc_file; _lookup_AD.

```
*****;
* Step 1. USER INPUT and parameters *;
*****;
LIBNAME indd "Your Input Library" ;
LIBNAME outdd "Your Output Library" ;
```

Replace the text in quotes with the path to the SAS library in which the input files are stored. Example:

```
*****;
* Step 1. USER INPUT and parameters *;
*****;
LIBNAME indd "C:\SAS\TRD" ;
LIBNAME outdd "C:\SAS\TRD" ;
```

The programs may be run interactively or batched.

Patient selection:

The programs were developed for patients:

1. Adults aged 18 to 64 years
2. At least two claims with depression diagnoses: 296.2x, 296.3x, 309.0, 300.4, 311.xx
3. No claims with any of the following diagnoses: 290.xx, 295.xx, 297.xx, 298.xx, 299.xx, 317.xx – 319.xx, 331.xx, 332.xx, 787.xx, 296.0, 296.1, 296.4, 296.5, 296.6, 296.7, 296.80, 296.81, 296.89
4. At least one claim for an antidepressant medication
5. At least two years of continuous enrollment following the first observed antidepressant medication claim

User inputs:

File Name: _pat_file

Patient file, including unique patient identifier and index date. Used to identify patient population meeting the patient selection criteria (above) and establish timeframe for each patient.

VARIABLE	FORMAT	DESCRIPTION
----------	--------	-------------

pat_id	6.	Unique patient identifier
index_date	mmddy10.	Date of first observed antidepressant prescription fill for the patient

File Name: `_ndc_file`

Prescription drug file, including unique patient identifier, National Drug Codes (NDC), daily dose, and service date for each prescription fill. Used to evaluate prescription drug use.

VARIABLE	FORMAT	DESCRIPTION
pat_id	6.	Unique patient number
NDCnum	\$11.	11-digit NDC with no hyphens or spaces (character)
Ddose	8.	Numeric variable for daily drug dose of prescription filled. Typically, this would be calculated using the strength of the pill, the number of pills dispensed, and the days the prescription is supposed to supply. For example, if 60 pills of 10 mg each are dispensed for a 30-day prescription, the daily dose would be 20 mg (= 10 mg/pill x 60 pills/day /30 days). Depending on your data source, you may or may not need to calculate this value. No units are required (i.e., enter 20, not 20 mg).
Daysupp	8.	Numeric variable for days supplied of prescription filled
Svcdate	mmddy10.	Date of prescription fill

File Name: `_lookup_AD.sas7bdat`

Lookup table for antidepressant medications, including NDC and generic name. Used to identify antidepressant medications, which are then evaluated in the prescription drug file (`_ndc_file`).

VARIABLE	FORMAT	DESCRIPTION
NDCnum	\$11.	11-digit NDC with no hyphens or spaces (character)
GenNme	\$50.	Generic name of drug. Any way of distinguishing different medications (e.g., a numeric value indicating active ingredient) will serve the purpose, if the variable is renamed GenNme and formatted as \$50.

MGH_AD.sas output:

File Name: `MGH_ANTIDEPRESSANT`

Patient-level file with each patient's MGH-AD score and its components

VARIABLE	FORMAT	DESCRIPTION
pat_id	6.	Unique patient identifier
MGH_AD	4.	MGH-AD Score
adequate	3.	Count of adequate antidepressant trials
titration	3.	Count of trials optimized with upward dose titration
duration	3.	Count of trials optimized with extended duration

References:

1. National Institute of Mental Health. Odds of Beating Depression Diminish as Additional Treatment Strategies are Needed. November 2006. Retrieved from: <http://www.nimh.nih.gov/science-news/2006/odds-of-beating-depression-diminish-as-additional-treatment-strategies-are-needed.shtml>



2. Fava M. Diagnosis and definition of treatment-resistant depression. *Biological Psychiatry* 2003; 53: 649-659.
3. Gibson TB, Jing Y, Bagalman JE, Kim E, Burton WN, Carls GS, Tran QV, Pikalov A, Goetzel RZ. Identification of treatment resistant depression in claims data: a comparison of methods. (PMH79) Poster Presentation ISPOR 14th Annual International Meeting, Orlando, FL. May 20, 2009.
4. Gibson TB, Jing Y, Carls GS, Kim E, Bagalman JE, Burton WN, Tran QV, Pikalov A, Goetzel RZ. The cost burden of treatment resistance in patients with depression. (PMH26) Poster Presentation ISPOR 14th Annual International Meeting. Orlando, FL, May 20, 2009.

The SAS code to compute MGH-AD begins on the next page.



```

/*****
/* Copyright 2009 Thomson Reuters. All rights reserved. */
/*****
/* Program name: mgh_antidepressant.sas */
/* ----- */
/* Project : */
/* Date Written : */
/* Description : Assign drug use scores during specified period. */
/* Input Files : Prescription drug claim files in specified format. */
/* Output Files : mgh_antidepressant */
/*****

*****;
* Step 1. USER INPUT and parameters *;
*****;
LIBNAME indd "Your Input Library" ;
LIBNAME outdd "Your output Library" ;

%LET patients=indd._pat_file; * PAT_ID, INDEX_DATE are expected *;

%LET ndc_clms=indd._ndc_file; * PAT_ID,SVCDATE,NDCNUM,DDOSE, and DAYSSUP
are expected *;
* DDOSE - daily drug dose per script. *;

* the following statement is used to filter out appropriate claims. *;
%LET select=(index_date LE svcdate LE (index_date + 730 -1 )) ;

%LET dsout=outdd.mgh_antidepressant;

*****;
* External look up tables. Prepared by the client. *;
*****;
%LET lookup=indd._lookup;
%LET antidepr =&lookup._ad ; * NDCNUM and GENNME (generic name) are
expected *;

*****;
* End of user input. No editing after this point. *;
*****;
OPTIONS nosource2 ;
*-----*;
* Sort Drug *;
*-----*;

PROC SORT DATA=&ndc_clms.; BY pat_id;run;

*-----*;
* Patient list *;
*-----*;
PROC SORT DATA=&patients. OUT=pat_list NODUPKEY;
BY pat_id;

```

```

RUN;

*-----*;
* Prefilter appropriate claims. *;
*-----*;

DATA rx ;
    MERGE
        &ndc_clms.
        pat_list(IN=idx)
    ;
    BY pat_id ;
    IF idx ;
    IF &select. ;
RUN ;

*-----*;
* Prepare look up tables and SAS formats. *;
*-----*;

*-----*;
* Antidepressants. Grouping *;
*-----*;

PROC SORT DATA=&antidepr.(KEEP=genme )
    OUT=gname
    NODUPKEY ;
    BY genme ;
RUN ;

DATA gname1 ;
    SET gname ;
    BY genme ;
    ad_grp + 1 ;
    CALL SYMPUT('ad_cnt', TRIM(LEFT(ad_grp)));
RUN ;
%LET ad_cnt=%TRIM(%LEFT(&ad_cnt.));

PROC SORT DATA=&antidepr. OUT=gname2 ;
    BY genme ;
RUN ;
DATA ad_grp ;
    MERGE
        gname2
        gname1
    ;
    BY genme ;
RUN ;

PROC SORT DATA=ad_grp NODUPKEY;
    BY ndcnum ;
RUN ;

DATA ad ;
    LENGTH end $11 hlo $1 label $6;
    SET ad_grp(KEEP=ndcnum ad_grp) end=eof;
    RETAIN fmtname "$AD_GRP" hlo " ";
    start = ndcnum;
    
```



```
end = ndcnum ;
label = PUT(ad_grp, 6.0) ;
OUTPUT ;
IF eof THEN DO;
  start = "other";
  label = " ";
  hlo = "0";
OUTPUT ;
END ;
RUN ;

PROC FORMAT cntlin=ad ;
RUN ;

%MACRO mgh_ad(rxin, dsout);
*-----*;
* Create Rx based flags.
*-----*;
DATA _rx ;
  LENGTH
    ad_grp 3
    record_type $16
  ;
SET &rxin. ;
*-----*;
IF PUT(ndcnum, $ad_grpf.) > ' ' THEN DO ;
  record_type = 'MGH' ;
  ad_grp = PUT(ndcnum, $ad_grpf.)*1 ;
  ddose = MAX(1, ddose) ;
  daysupp = MAX(1, daysupp) ;
OUTPUT ;
END ;
RUN;

DATA mgh_rx ;
  LENGTH
    ADEQUATE
    TITRATION
    DURATION
  3;
LABEL
  pat_id = "Enrollee ID"
  ADEQUATE = "count of adequate trials (two fills)"
  TITRATION = "count of optimized trials (dose increase)"
  DURATION= "Count of unique agents in THERCLS=69 with at least 3 fills"
  ;
KEEP pat_id
  ADEQUATE
  TITRATION
  DURATION
  ;

SET _rx (WHERE=(record_type = 'MGH'));
BY pat_id svcdte ;
```



```
ARRAY fills_ (&ad_cnt.) _temporary_ ;
ARRAY scrpts_ (&ad_cnt.) _temporary_ ;
ARRAY doses_ (&ad_cnt.) _temporary_ ;
ARRAY incrs_ (&ad_cnt.) _temporary_ ;

* there are &ad_cnt. distinct drug groups *;
IF FIRST.pat_id THEN DO K=1 TO &ad_cnt. ;
  fills_(K) = 0 ;
  scrpts_(K) = 0 ;
  doses_(K) = . ;
  incrs_(K) = 0 ;
END ;

* count re-fills by unique drug groups *;
fills_(ad_grp) + 1 ;

* count adjusted re-fills by unique drug groups *;
IF daysupp LE 60 THEN scrpts_(ad_grp) + 1 ;
ELSE IF daysupp GT 60 THEN scrpts_(ad_grp) + 2 ;

* test for dose increase by unique drug groups *;
IF ddose > 0 THEN DO ;
  IF (doses_(ad_grp) = .) THEN
    doses_(ad_grp) = ddose ;
  ELSE IF (doses_(ad_grp) > .) AND (ddose > doses_(ad_grp)) THEN DO ;
    doses_(ad_grp) = ddose ;
    incrs_(ad_grp) = 1 ;
  END ;
END ;

IF LAST.pat_id THEN DO ;
  ADEQUATE = 0 ;
  TITRATION = 0 ;
  DURATION = 0 ;

  DO K=1 TO &ad_cnt. ;
    IF fills_(K) > 1 THEN ADEQUATE = ADEQUATE + 1 ;

    IF scrpts_(K) > 2 THEN DURATION = DURATION + 1 ;
    TITRATION = TITRATION + incrs_(K) ;
  END ;
  OUTPUT ;
END ;
RUN ;

DATA &dsout. ;
LENGTH MGH_AD 4 ;
LABEL
  pat_id      = "Enrollee ID"
  ADEQUATE    = "count of adequate trials (two fills)"
  TITRATION   = "count of optimized trials (dose increase)"
  DURATION    = "Count of unique agents with at least 3 fills"
  MGH_AD      = "Antidepressant-only MGH"
```



```

;

KEEP pat_id
  ADEQUATE
  TITRATION
  DURATION
  MGH_AD
;

MERGE
  mgh_rx
  pat_list(IN=idx KEEP=pat_id)
;

BY pat_id ;
IF idx ;

*-----*
  MGH_AD = ADEQUATE + 0.5*(TITRATION + DURATION);
*-----*

RUN;

%MEND mgh_ad ;

%mg_h_ad(rx, &dsout.);
```